
1. Description  
 
This file includes the BUGS code for fitting the reduced model and the full model to the data. 

The reduced model does not explicitly include resource limitation as opposed to the full model. 

In the reduced model, the density-independent growth rate of the ith phytoplankton functional 

type from one week to the next is denoted as 𝑟 . The full model incorporates resource limitation 

by desribing the realized density-independent growth rate of the ith phytoplankton functional 

type from week 𝑤 − 1 to 𝑤 as 𝑟 , = 𝜇  𝑓 (𝑤), where 𝜇 > 0 is the maximum net growth rate of 

the ith functional type from one week to the next at average environmental conditions 

(temperature and salinity) and optimal resource combination, and 

𝑓 (𝑤) = 𝑚𝑖𝑛
𝑃𝐴𝑅

𝑃𝐴𝑅 + 𝐾𝐸
,

𝑁𝑖𝑡

𝑁𝑖𝑡 + 𝐾𝑁
,

𝑆𝑖𝑙

𝑆𝑖𝑙 + 𝐾𝑆
 

 

 

where 𝐾𝐸 > 0, 𝐾𝑁 > 0 and 𝐾𝑆  > 0 denote respectively the irradiance, nitrogen and silicate 

half-saturation constants representing the level of each resource at which 𝑟 , = 𝜇 /2, provided 

that none of the other resources is limiting.  

2. BUGS Code for the reduced and the full (extended model) 
 

Nobs and Npop denote respectively the length of the time series (number of weeks) and the 

number of populations (functional groups) under consideration.  Note that in BUGS, the normal 

distribution is parameterized with the mean and the precision (inverse variance). 

 
2.1. Reduced model 
 
model{  

##### DYNAMICAL MODEL ##################################### 

for(t in 2:Nobs){ 

# Incorporating "residual" environmental stochasticity  

w[t,1:Npop]~dmnorm(m[t,1:Npop], Tau[,])  

for(i in 1: Npop){ 

# Adding biotic interactions and environmental forcing through environmental 

variables 

m[t,i]<- y[t-1,i]+r[i]*(1+ inprod(alpha[i,], y[t-1,])/k[i]) + 

beta[i,1]*temperature[t] + beta[i,2]*salinity[t] 

# Incorporating demographic stochasticity 

y[t,i]~dnorm(w[t,i], Dem.prec[t,i]) 

ypred[t,i]~dnorm(w[t,i], Dem.prec[t,i]) 

eps[t,i]<-(y[t,i]-ypred[t,i]) 

Dem.var[t,i]<-1/Dem.prec[t,i] 

# Scaling of demographic variance with the (population size) FG biomass 

Dem.prec[t,i]<-tau.d[i]*exp(y[t,i]) 

 }## end i 

  }## end t 

## r[i] is the intrinsic growth rate for FG i  

## alpha[i,j] effect of Functional Group (FG) j on the growth rate of  FG i  

## alpha[i,i] intraspecific effect ####### 

## Specifying Priors ################################# 

for(s in 1:Npop){ 

         for(v in 1:Nvar){ 

           beta[s,v]~dnorm(0,tau.b[s,v]) 

           tau.b[s,v]<-v.b[s,v] 

           v.b[s,v]~dexp(lambda.b) 

      } 

   } 

alpha[Npop, Npop]<- -1 

for(i in 1:Npop-1){  

 alpha[i,i]<- -1 

    for(j in i+1:Npop){   

      alpha[i,j]~dnorm(0, tau.a[i,j]) 

       alpha[j,i]~dnorm(0, tau.a[j,i]) 

       tau.a[i,j]<-1/  v.a[i,j] 

       tau.a[j,i]<-1/  v.a[j,i] 

       v.a[i,j]~dexp(lambda.a) 

       v.a[j,i]~dexp(lambda.a) 

    }  

 } 

Tau[1:Npop,1:Npop] ~ dwish(R[ , ], Npop) 

 # Environmental covariance matrix 

 sigma[1:Npop,1:Npop]<-inverse(Tau[,]) 

 #  Cross-species environmental correlations 

 for(i in 1:Npop){  

    for(j in 1:Npop){  

       rho[i,j]<-sigma[i,j]/sqrt(sigma[i,i]*sigma[j,j]) 

      alpha.square[i,j]<-pow(alpha[i,j],2) 

   } 

 } 

 for(i in 1:Npop){ 

   r[i]~dgamma(6,4) 

   k[i]~dgamma(6,4) 

   h[i]<-r[i]/k[i] 

   tau.d[i]~dgamma(1,1) 

   demVar[i]<-1/tau.d[i]    

   ## Variance due to Env. forcing and intra/inter-spec. interactions  

   stress1[i]<-(beta[i,1]*beta[i,1]) 

   stress2[i]<-(beta[i,2]*beta[i,2]) 

   intra[i]<-pow((r[i]/k[i]),2)*svar[i] 

   ## svar[i] is the variance of y[,i]  

   totinteract[i]<-pow((r[i]/k[i]),2)*inprod(alpha.square[i,],svar[]) 

   inter[i]<-totinteract[i]-intra[i] 

   ## Total variance for population i (excluding demographic stochasticity)  

   totvar[i]<-  totinteract[i]+ stress1[i]+ stress2[i] + sigma[i,i] 

   ## contributions of environmental and biotic factors 

   cenv[i]<-(stress1[i]+ stress2[i] + sigma[i,i])/ totvar[i] 

   cbio[i]<-totinteract[i] / totvar[i] 

   cintra[i]<-intra[i]/totvar[i] 

   cinter[i]<-inter[i]/totvar[i] 

  } 

  lambda.b~dgamma(10,1) 

  lambda.a~dgamma(10,1) 

 }## end model 

 
############################################################################# 
 

2.2. Full model 
 
model{  

for(t in 2:Nobs){ 

w[t,1:Npop]~dmnorm(m[t,1: Npop], Tau[,])  

for(i in 1: Npop){ 

# Including biotic interactions and environmental forcing  

m[t,i]<- y[t-1,i]+g[t,i] 

g[t,i]<-r[t,i]*(1+ inprod(alpha[i,], y[t-1,])/k[i]) + 

beta[i,1]*temperature[t] + beta[i,2]*salinity[t] 

r[t,i]<-mu[i]*min(nitrogen[t]/(KN[i]+nitrogen[t]), min(PAR[t]/(KE[i]+PAR[t]), 

silicate[t]/(KS[i]+silicate[t]) ) ) 

 h[t,i]<-r[t,i]/k[i] 

# Incorporating demographic stochasticity 

y[t,i]~dnorm(w[t,i], Dem.prec[t,i]) 

Dem.var[t,i]<-1/Dem.prec[t,i] 

# Inverse scaling of demographic variance with the (population size) FG 

biomass 

Dem.prec[t,i]<-tau.d[i]*exp(y[t,i]) 

 }## end i 

  }## end t 

## r[i] is the intrinsic growth rate for FG i at time t #### 

# alpha[i,j] effect of FG j on the growth rate of  FG i  

## alpha[i,i] intraspecific effect ####### 

##### Specifying Priors ################################# 

for(s in 1:Npop){ 

         for(v in 1:Nvar){ 

           beta[s,v]~dnorm(0,tau.b[s,v]) 

           tau.b[s,v]<-v.b[s,v] 

          v.b[s,v]~dexp(lambda.b) 

      } 

   } 

alpha[Npop, Npop]<- -1 

for(i in 1:Npop-1){  

 alpha[i,i]<- -1 

  for(j in i+1:Npop){   

      alpha[i,j]~dnorm(0, tau.a[i,j]) 

      alpha[j,i]~dnorm(0, tau.a[j,i]) 

      tau.a[i,j]<-1/v.a[i,j] 

      tau.a[j,i]<-1/v.a[j,i] 

      v.a[i,j]~dexp(lambda.a) 

      v.a[j,i]~dexp(lambda.a) 

    }  

 } 

Tau[1:Npop,1:Npop] ~ dwish(R[ , ], Npop) 

 # Environmental covariance matrix 

 sigma[1:Npop,1:Npop]<-inverse(Tau[,]) 

 #  Cross-species environmental correlations 

 for(i in 1:Npop){  

    for(j in 1:Npop){  

       rho[i,j]<-sigma[i,j]/sqrt(sigma[i,i]*sigma[j,j]) 

   } 

 } 

for(i in 1:Npop){ 

 mu[i]~dgamma(6,4) 

   k[i]~dgamma(6,4) 

  # h[i]<-r[i]/k[i] 

   tau.d[i]~dgamma(1,1) 

   demVar[i]<-1/tau.d[i]    

   KN[i]~dnorm(0,100)I(0,) 

   KS[i]~dnorm(0,tauS[i])I(0,) 

   KE[i]~dnorm(15,0.01)I(0,) 

   tauS[i]<-1/varS[i] 

   varS[i]<-gam[i]*1 +(1-gam[i])*0.001 

   #varS[i]~dexp(1) 

   gam[i]~dbern(0.25) 

   r[1,i]<-mu[i]*min(nitrogen[1]/(KN[i]+nitrogen[1]),   

min(PAR[1]/(KE[i]+PAR[1]), silicate[1]/(KS[i]+silicate[1]) ) ) 

 } 

  lambda.b~dgamma(10,1) 

  lambda.a~dgamma(10,1) 

  }## end model 

############################################################################# 


